# Thunar-vfs, the virtual filesystem.

pkgname=thunar-vfs
pkgver=1.2.0
pkgrel=1
arch=('ppc')
pkgdesc="The virtual filesystem for Thunar"
url="http://git.xfce.org/xfce/thunar-vfs/"
license=('GPL2')
depends=('libx11' 'exo>=0.6.0' 'glib2' 'gtk2' 'gdk-pixbuf2' \
				'libxfce4util>=4.8.0' 'libpng' 'dbus-glib' 'startup-notification' \
				'libjpeg')
makedepends=('intltool' 'pkg-config' 'perl' 'gettext')
source=(http://archive.xfce.org/src/xfce/${pkgname}/1.2/${pkgname}-${pkgver}.tar.bz2)
md5sums=('1fbc55af8ed98174f5c3c7f8daec10cc')

build() {
	cd "${srcdir}/${pkgname}-${pkgver}"

	sh configure --prefix=/usr \
		--sysconfdir=/etc \
		--libexecdir=/usr/lib/xfce4 \
		--localstatedir=/var \
		--disable-static \
		--enable-nls \
		--enable-dbus \
		--disable-gnome-thumbnailers \
		--enable-startup-notification \
		--disable-gtk-doc-html \
		--without-html-dir \
		--with-volume-manager=auto

	make
}

package() {
	cd "${srcdir}/${pkgname}-${pkgver}"

	make DESTDIR=${pkgdir} install
}
